
Script 1
Description:
The script is written in OriginLab's programming language LabTalk. First, it calculates the relative standard error of the intensity signals by dividing the intensity standard error by the intensity. The first intensity point with a relative standard error of >= 0.1 corresponds to the laser switching to the first wavelength of the scan after the LabView program was started. The script notes the row index of this data point, and it also finds the next intensity point when the laser steps to the second wavelength of the scan. Using these two indices and the fact that there are 3 seconds between each data point, the script calculates the number of seconds per step. Then, from the first index, it figures out how many rows at the beginning of the scan to delete (those corresponding to data points before the LabView scan started and while the laser stabilizes at the first wavelength). It then deletes all of the data points during the scan while the laser is stepping to the next wavelength and stabilizing (8 data points, or 24 seconds). The data points left are the ones used to calculate the two-photon spectra for the sample sitting in the 'Sample' holder and the sample sitting in the 'Reference' holder. This is also the data that is plotted for the visual check that the script worked (Figure 8 as an example).
The script then splits the data into two worksheets containing the 'Sample' data points and the 'Reference' data points. For each worksheet, the data points corresponding to each wavelength are averaged, resulting in one data point per wavelength. This is then copied into the final worksheet (named '2PA'), where the first column is populated with the laser wavelengths. The 'sample' and 'reference' data are found in the 'Sample raw data' and 'Reference raw data' columns, respectively. Columns D and E are reserved for copying and pasting the 'sample' and 'reference' correction functions, respectively. Columns F and G are set up to auto-calculate the correct two-photon spectral shapes after the correction functions have been pasted into D and E by dividing the raw data by the correction function. Or, if you are measuring the correction function of the setup with reference dyes, columns F and G will auto-calculate the correction functions after you paste the correct two-photon spectral shapes of the dyes into columns D and E.

wks.col1.width = 15;
wks.col1.lname$=Substitute(wks.col1.lname$,"",Title=)$;
wks.col1.comment$=Substitute(wks.col1.comment$,"",Comment=)$;
wks.col4.comment$="Standard Error";
com$=wks.col1.comment$;
csetvalue formula:=abs(D/C) col:=5 recalculate:=0;
wks.col5.comment$="Relative Standard Error";
step=4;
rse=5; //relative standard error column number
for (ii=1, found=0; found!=3;ii++)
	{
		if (col($(rse))[$(ii)] >= 0.1)
		{
			if (found==0) 
			{
				found=1;
				ind1=ii;
			};	
			else
				if (found==2)
				{
					found=3;
					ind2=ii;
				};
		};	
		else
			if (found==1) found=2;
		if (ii==100) break;
}
sec = (ind2-ind1)*3;
wks.col4.lname$=$(sec)-sec step;
wcopy 1! <new>!<new> ;
wks.name$ = cal;

title$=page.longname$;

// now get the box to pop up:
getnum (First wavelength:) first (Last wavelength:) last (Wavelength step:) step (Seconds per step:) sec (%(title$));


x=sec/3-8; //x is the number of 'good' data points for every wavelength.
//there are 3 seconds between each data point.
//it takes 24 seconds for the laser to stabilize fully.
//so, you divide the number of seconds by three and subtract the data when the //laser is stabilizing (8*3 = 24).

rws=ind1+6; //ind1 is the first index where the laser is seen to be unstable. //(std error is >=0.1)
//to get the number of rows to delete, you add 6 because that gives 1 data //point of wiggle room
//on the right edge of the wavelength measurement.


wks.deleteRows (1, rws);

wreducerows nrows:=8 skip:=x start:=x+1;

act$=page.name$;

plotxy 3 plot:=201;

win -a %(act$);

wcopy 2! <new>!<new>; // copying sheet 2

wreducerows nrows:=1 skip:=1 start:=1;

if (col(B)[1]$ == Sample) wks.name$ = Sample; else wks.name$ = Reference;

col(e)=ave(col(c),x/2);
wks.col5.comment$ = reduced; // labeling the column

wcopy 2! <new>!<new>; // copying sheet 2 again

wreducerows nrows:=1 skip:=1 start:=2;

if (col(B)[1]$ == Sample) wks.name$ = Sample; else wks.name$ = Reference;

col(e)=ave(col(c),x/2);

wks.col5.comment$ = reduced;

newsheet name:="2PA";

wks.nCols = wks.nCols + 5; // adding 5 columns

col(a) = {first:step:last};

page.active$ = Sample;

//To copy column 5 of selected worksheet to column B of 2PA:
colcopy irng:=(5:5) orng:=2PA!B;

page.active$ = Reference;

colcopy irng:=(5:5) orng:=2PA!C;
page.active$ = 2PA;

nsteps=((last-first)/step)+1;

wks.deleteRows($(nsteps+1),$(wks.nrows-nsteps));

wks.col1.lname$= "Wavelength";
wks.col1.unit$="nm";
wks.col1.comment$ = com$;
wks.col2.comment$ = "Sample raw data";
wks.col3.comment$ = "Reference raw data";
wks.col4.comment$ = "Paste sample correction";
wks.col5.comment$ = "Paste reference correction";
wks.col6.comment$ = "Sample corrected";
wks.col7.comment$ = "Reference corrected";


csetvalue formula:=col(B)/col(D) col:=6 recalculate:=1;
csetvalue formula:=col(C)/col(E) col:=7 recalculate:=1;

